

Perl Weekly Challenge - 003

2019-04-21 13:04 +00 • [Mark Senn](#)

Challenge #1

From [Perl Weekly Challenge - 003](#) retrieved on 2019-04-13 at 03:39 +00:

Create a script to generate 5-smooth numbers, whose prime divisors are less or equal to 5. They are also called Hamming/Regular/Ugly numbers. For more information, please check this [wikipedia](#).

That Wikipedia entry retrieved on 2019-04-13 at 03:44 +00 contained:

Regular numbers are numbers that evenly divide powers of 60 (or, equivalently powers of 30). As an example, $60^2 = 3600 = 48 \times 75$, so both 48 and 75 are divisors of a power of 60. Thus, they are *regular numbers*. Equivalently, they are the numbers whose only prime divisors are 2, 3, and 5.

The challenge provides no details regarding which 5-smooth numbers to generate or how to print them. I'm only going to generate the numbers

$$2^i \times 3^j \times 5^k \quad \text{for } i \in E, j \in E, \text{ and } k \in E$$

where $E = \{1, \dots, m\}$ and m is the maximum exponent.

The Perl 6 program:

```
# Perl Weekly Challenge - 003
# Challenge #1
#
# See
#   engineering.purdue.edu/~mark/pwc-003.pdf
# for more information.

# Run using Perl 6.
use v6;

# Maximum exponent.
my $m = 2;

# Construct the result.
my @r = ();
# No need to get fancy---nested loops
# will produce only the numbers needed.
for (1..$m) -> $i {
    for (1..$m) -> $j {
        for (1..$m) -> $k {
            @r.push(2**$i * 3**$j * 5**$k);
        }
    }
}

# Print the result.
@r.sort.join("\n").say;
```

Challenge #2


From [Perl Weekly Challenge - 003](#) retrieved on 2019-04-13 at 04:41 +00:

Create a script that generates Pascal Triangle. Accept number of rows from the command line. The Pascal Triangle should have at least 3 rows. For more information about Pascal Triangle, check this [wikipedia](#) page.

That Wikipedia entry retrieved on 2019-04-13 at 04:53 +00 contained:

The rows of Pascal's triangle are conventionally enumerated starting with row $n = 0$ at the top (the 0th row). The entries in each row are numbered from the left beginning with $k = 0$ and are usually staggered relative to the numbers in the adjacent rows. The triangle may be constructed in the following manner: In row 0 (the topmost row), there is a unique nonzero entry 1. Each entry of each subsequent row is constructed by adding the number above and to the left with the number above and to the right, treating blank entries as 0. For example, the initial number in the first (or any other) row is 1 (the sum of 0 and 1), whereas the numbers 1 and 3 in the third row are added to produce the number 4 in the fourth row.

[The following picture is based on this [TikZ code](#). —mark]



The entry in the n th row and k th column of Pascal's triangle is denoted $\binom{n}{k}$. For example, the unique nonzero entry in the topmost row is $\binom{0}{0} = 1$. With this notation, the construction of the previous paragraph may be written as follows:

$$\binom{n}{k} = \binom{n-1}{k-1} + \binom{n-1}{k}$$

The challenge provides no details regarding how the rows should be printed. I decided to print the numbers in the shape of a triangle. Printing the, for example, 1 in row 0 centered horizontally relative to the 20 in row 6 is beyond

the scope of this challenge. Some illustrations with vertical lines to show alignment:

Can't do: $\left| \begin{array}{c} 1 \\ 20 \end{array} \right|$ Can do: $\left| \begin{array}{c} 1 \\ 20 \end{array} \right|$

To format the triangle, the width of the largest element of the triangle is needed. For a table with y rows, the largest element will be in row y , column $x = \lfloor y/2 \rfloor$. In the $y = 6$ row table above, $x = \lfloor 6/2 \rfloor = 3$, and the largest element is 20. Let w be the number of digits in the largest element.

Start each row with $(y - n) \times (1 + w)$ spaces followed by a 1. Each subsequent column should be

- prefixed by a space
- followed by w spaces (for the missing number in that column),
- followed by another space
- followed by the column value right-justified in a field w characters wide

Here is a picture for a $y = 6$ triangle:

```

      1
     1 1
    1 2 1
   1 3 3 1
  1 4 6 4 1
 1 5 10 10 5 1
1 6 15 20 15 6 1

```

See the next page for the Perl 6 program.

The Perl 6 program:

```
# Perl Weekly Challenge - 003
# Challenge #2
#
# See
#   engineering.purdue.edu/~mark/pwc-003.pdf
# for more information.

# Run using Perl 6.
use v6;

sub MAIN(Int $y where * >= 3)
{
    my $x = ($y/2).floor;

    my $max-value = binom($y, $x);
    my $w = $max-value.chars;

    for (0..$y) -> $n
    {
        # Start each row with ($y-$n) * (1+$w) spaces followed by a 1.
        print ' ' x ($y-$n) * (1+$w);
        print '1';

        # Each subsequent column should be
        #   o prefixed by a space
        #   o followed by $w spaces (for the missing number in that column)
        #   o followed by another space
        #   o followed by the column value right-justified in a field $w characters wide
        for (1..$n) -> $k
        {
            print ' ';
            print ' ' x $w;
            print ' ';
            printf "%{$w}d", binom($n,$k);
        }
        print "\n";
    }
}

sub binom($n, $k)
{
    fact($n) / (fact($k) * fact($n-$k))
}

sub fact($n)
{
    ($n == 0)
    ?? 1
    !! [*] (1..$n)
}

sub USAGE()
{
    note "usage: perl6 {$*PROGRAM-NAME} non-negative_integer_of_at_least_three";
}
}
```